Вывод метрики Шварцшильда из вариационного принципа Гильберта с помощью пакета wxMaxima

Шитова А. М.

13 июля 2014 г.

Аннотация

В данной статье приведено подробное описание алгоритма Виктора Тота (Victor Toth) по выводу метрики Шварцшильда на основе вариационного принципа Гильберта. Приведены физические основания данного вывода.

1 Пакеты ctensor и itensor в WxMaxima

Для тензорных вычислений в Maxima — открытой компьютерной алгебраической системе — существуют три пакета: **atensor**, **ctensor** и **itensor**.

Пакет **atensor** [1] используется для алгебраических вычислений в различных алгебрах (например, в алгебре кватернионов). Два других пакета предназначены для символьных тензорных преобразований двух различных типов: по компонентам (пакет **ctensor** [2]) и с помощью индексов (пакет **itensor** [3]). В первом случае тензоры представлены как массивы или матрицы, тогда как во втором под тензорами понимаются функции их ковариантных, контрвариантных и производных индексов. Таким образом, в пакете **ctensor** такие тензорные операции как свёртка или ковариантное дифференцирование производятся непосредственно с помощью компонент, а в пакете **itensor** — с помощью индексов. Естественно, каждый из этих двух подходов имеет свои достоинства и недостатки. Индексные выражения, полученные с помощью пакета **itensor** могут быть переведены в координатное представление пакета **ctensor** с помощью функции **ic_convert**.

Пакет **itensor** позволяет проводить вычисления в формализме Лагранжа, а именно получать уравнения Лагранжа-Эйлера в индексной форме. Пример использования пакета **itensor** — программа Виктора Тота [4] (см. также [5]) для получения с помощью действия Эйнштейна-Гильберта тензора Эйнштейна (уравнений Фридмана) и сферически-симметричного статичного решения Шварцшильда. Хотя пример сопровождается поясняющими комментариями, на наш взгляд, неискушенному читателю следует более подробно пояснить основные моменты, как с точки зрения физики, так и с точки зрения программирования.

В качестве примера использования пакета **ctensor** приведём также текст программы для вывода метрики Шварцшильда и Райсснера-Нордстрема непосредственно из уравнений Эйнштейна (см. Приложение). Отметим, что для студентов методически более полезно проделать весь вывод, в том числе получение компонент тензора Эйнштейна, «руками», так как, эти метрики — немногочисленные примеры точных решений.

2 Вариационный принцип Гильберта

Известно, что Гильберт получил уравнения гравитации раньше, чем Эйнштейн опубликовал свою статью. Тем не менее, в начале XXI века появились работы, стремившиеся опровергнуть устоявшуюся точку зрения на историю этого фундаментального открытия. Однако, как было подробно показано в статьях [6, 7], альтернативного мнения на этот счет быть не может. История ясно говорит, что зарождение ОТО — плод работы трёх людей: Эйнштейна, Гроссмана и Гильберта, и умаливать достижения кого-то из них некорректно и неэтично. Хотя определенные тайны всё же остались, в частности, неизвестным была вырезана часть записей из гранок статьи Гильберта. Оставим эту проблему историкам науки и перейдём непосредственно к рассуждениям Гильберта.

Первая аксиома Гильберта [8] формулируется следующим образом: «Закон физического события определяется мировой функцией H (*лагранжианом*), аргументы которой таковы:

$$g_{\mu\nu}, g_{\mu\nu,l} = \frac{\partial g_{\mu\nu}}{\partial x^l}, g_{\mu\nu,lk} = \frac{\partial^2 g_{\mu\nu}}{\partial x^l \partial x^k}, q_s, q_{s,l} = \frac{\partial q_s}{\partial x^l}, (l, m, k = 1, 2, 3, 4), \tag{1}$$

причем вариация интеграла $\int H\sqrt{|g|}dx_1 dx_2 dx_3 dx_4$ обращается в нуль для каждого из 14 потенциалов $g_{\mu\nu}$ и q_s». Под x_s понимаются наиболее общие пространственно-временные координаты (у Гильберта w), $g_{\mu\nu}$ — введенные Эйнштейном гравитационные потенциалы, q_s — электродинамическтие потенциалы. Можно аналогично использовать и аргументы с контрвариантными индексами.

Для того чтобы в уравнения гравитации входили лишь вторые производные потенциалов $g^{\mu\nu}$, функция H должна иметь вид $H = R + L$, где $R = R_{\mu\nu}g^{\mu\nu}$ — скалярная кривизна четырехмерного многообразия ($R_{\mu\nu}$ — тензор Римана), а L — функция только переменных $g^{\mu\nu}$, $g^{\mu\nu}_{,l}$, q_s и q_{sk}. Для простоты Гильберт также предположил, что L не зависит от $g^{\mu\nu}_{,l}$. При вычислении вариации действия $\delta S = \int H\sqrt{g}d^4x$ все члены, связанные с изменением q, необходимо занулить [9] (что соответствует введению уравнений движения). Из первой аксиомы Гильберта при варьировании по остальным 10 гравитационным потенциалам $g^{\mu\nu}$, получаем:

$$\delta S = \int \left(\frac{\partial \sqrt{-g} H}{\partial g^{ik}} \delta g^{ik} + \frac{\partial \sqrt{-g} H}{\partial g^{ik}_{,l}} \delta \frac{\partial g^{ik}}{\partial x^l} + \frac{\partial \sqrt{-g} H}{\partial g^{ik}_{,lm}} \delta g^{ik}_{,lm} \right), \qquad (2)$$

откуда по теореме Гаусса, полагая на границах интегрирования $\delta g^{ik} = 0$, следуют 10 дифференциальных уравнений Лагранжа, которые Гильберт назвал «основными уравнениями гравитации»[8]:

$$\frac{\partial \sqrt{g} H}{\partial g^{\mu\nu}} - \sum_k \frac{\partial}{\partial w_k} \frac{\partial \sqrt{g} H}{\partial g^{\mu\nu}_{,k}} + \sum_{k,l} \frac{\partial^2}{\partial w_k \partial w_l} \frac{\partial \sqrt{g} H}{\partial g^{\mu\nu}_{,kl}} = 0, \quad (\mu, \nu = 1, 2, 3, 4). \qquad (3)$$

Легко проверить, что вариационный принцип Гильберта приводит к уравнениями Эйнштена. Действительно, вычислим вариацию действия $\delta \int R \sqrt{-g} d^4 x$:

$$\delta S = \delta \int d^4 x \sqrt{-g} R = \delta \int d^4 x \sqrt{-g} g^{\mu\nu} R_{\mu\nu} =$$
$$= \int d^4 x \sqrt{-g} \left(\delta g^{\mu\nu} R_{\mu\nu} + \frac{\delta \sqrt{-g}}{\sqrt{-g}} R + g^{\mu\nu} \delta R_{\mu\nu} \right). \qquad (4)$$

Рассмотрим вариацию $\delta \sqrt{-g}$. Прежде всего, докажем тождество: $\delta \det A = \delta A \mathrm{Sp}(A^{-1} \delta A)$, где A — произвольная матрица. Для этого рассмотрим вариацию логарифма (см., например, [10, 11]): $\delta \ln \det A$:

$$\delta \ln \det A = \ln \det(I + A^{-1} \delta A) = \ln(1 + \mathrm{Sp} A^{-1} \delta A) = \mathrm{Sp} A^{-1} \delta A. \qquad (5)$$

Далее, варьируя единичную матрицу, «хитро записанную» в виде $g_{\alpha\beta} g^{\alpha\sigma}$, получаем: $\delta g_{\alpha\beta} g^{\alpha\sigma} + g_{\alpha\beta} \delta g^{\alpha\sigma} = 0$ (аналогично можно показать, что $\delta g^{\mu\nu} = -g^{\mu\alpha} g^{\nu\beta} \delta g_{\alpha\beta}$). Следовательно, $\delta g_{\alpha\beta} g^{\alpha\sigma} = -g_{\alpha\beta} \delta g^{\alpha\sigma}$.

Таким образом,

$$\delta \sqrt{-g} = -\frac{1}{2\sqrt{-g}} \delta g = -\frac{1}{2\sqrt{-g}} g g^{\alpha\beta} \delta g_{\alpha\beta} = -\frac{1}{2} \sqrt{-g} g_{\alpha\beta} \delta g^{\alpha\beta}. \qquad (6)$$

Третье слагаемое в уравнении (4) не даст вклада в конечные уравнения, так как сводится к интегралу от полной дивергенции (см., например, [9, 10]). Таким образом,

$$\delta S = \delta \int d^4 x \sqrt{-g} R = \int d^4 x \sqrt{-g} \left(R_{\mu\nu} - \frac{1}{2} R g_{\mu\nu} \right) \delta g^{\mu\nu}. \qquad (7)$$

В скобках образовался тензор Эйнштейна. Из принципа наименьшего действия для суммы вариаций $\delta S + \delta S_m$, где S_m – действие материи, можно получить уравнения Эйнштейна в их традиционной форме. В пустом пространстве, как следует из уравнения (7), ввиду произвольности вариации $\delta g^{\mu\nu}$, $R_{\mu\nu} - \frac{1}{2} R g_{\mu\nu} = 0$.

3 Алгоритм Виктора Тота для вывода метрики Шварцшильда из вариационного принципа Гильберта

Решение Шварцшильда — исторически первое точное решение уравнения Эйнштейна — описывает геометрию пространства-времени вокруг незаряженного сферически-симметричного источника. Метрику Шварцшильда можно вывести прямо из вариационного принципа Гильберта и не переходя к уравнениям Эйнштейна.

Разберём непосредственно действия алгоритма [5]. Поясним пошагово, что делает каждая из описанных процедур.

*Для начала необходимо подключить пакеты для тензорных вычислений **ctensor** и **itensor**.*

```
(%i1)  if get('ctensor,'version)=false then load(ctensor);
```

%i1 Подключает пакет ctensor (если ещё не подключен) для тензорных вычислений

```
(%i2)  if get('itensor,'version)=false then load(itensor);
```

%i2 Подключает пакет itensor (если ещё не подключен) для символьных вычислений с тензорными индексами.

Далее мы должны сконструировать тензор Римана. Для этого В. Тот определяет вспомогательный симметричный тензор.

```
(%i3)  remsym(g,2,0);
       remsym(g,0,2);
       remsym(gg,2,0);
       remsym(gg,0,2);
       remcomps(gg);
       imetric(gg);
```

%i3 Убирает все симметрии из тензоров **g** и **gg** с 2мя ковариантными индексами; аналогично с 2мя контрвариантными. Очищает компонентные значения тензора **gg**. Объявляет тензор **gg** метрикой.

```
(%i9)  icurvature([a,b,c],[e])*gg([d,e],[])$
```

%i9 Умножает тензор кривизны (Римана) с компонентами R^{e}_{abc} на метрический тензор gg_{de}. (*Осуществляем переход от смешанных компонент тензора кривизны к ковариантным.*)

```
(%i10) contract(rename(expand(%)))$
```

%i10 Сворачивает по повторяющимся индексам e предыдущее выражение, предварительно осуществив упрощение и заменив немые индексы (*Функция **expand** «улучшает» работу функции **contract***).

```
(%i11) %,ichr2$
```

%i11 Подставляет в ковариантный тензор кривизны явные выражения для символов Кристоффеля.

ichr2 — Символы Кристоффеля второго рода $ichr2^{k}_{ij} = g^{ks}(g_{is,j} + g_{js,i} - g_{ij,s})/2$, где запятой обозначена частная производная $g_{is,j} =$

$\partial g_{is}/\partial x^j$.

```
(%i12) contract(rename(expand(%)))$
```

%i12,14,18 то же, что и %i10 для %i11.

```
(%i13) canform(%)$
```

%i13 Упрощает выражение, сворачивая по немым индексам.

```
(%i14) contract(rename(expand(%)))$
(%i15) components(gg([a,b],[]),kdels([a,b],[u,v])*g([u,v],[])/2);
```

%i15 Присваивает компонентам метрического тензора **gg** половину значения $\delta^{u,v}_{a,b} \cdot g_{u,v}$, где δ — симметричная свёртка с символами Кронекера: $\delta^{u,v}_{a,b} = \delta^u_a \delta^v_b - \delta^v_a \delta^u_b$. (*Симметризует метрический тензор.*)

```
(%i16) components(gg([],[a,b]),kdels([u,v],[a,b])*g([],[u,v])/2);
```

%i16 То же с верхними индексами.

```
(%i17) %th(4),gg$
```

%i17 Подставляет в выражение %i14 метрический тензор.

```
(%i18) contract(rename(expand(%)))$
(%i19) contract(rename(expand(%)))$
```

%i19 Упрощает.

```
(%i20) imetric(g);
```

%i20 Объявляет тензор g метрикой.

```
(%i21) contract(rename(expand(%th(2))))$
(%i22) remcomps(R);
```

%i22 Очищает компоненты **R**.

```
(%i23)  components(R([a,b,c,d],[]),%th(2));
(%i24) g([],[a,b])*R([a,b,c,d])*g([],[c,d])$
```

%i24 Строит скалярную кривизну $g^{ab} R_{abcd} g^{cd}$ и упрощает.

```
(%i25) contract(rename(canform(%)))$
(%i26) contract(rename(canform(%)))$
(%i27) components(R([],[]),%);
```

(%o27) *done*

```
(%i28) decsym(g,2,0,[sym(all)],[]);
```

%i28 Объявляет симметрию для 2х ковариантных координат тензора g ($g_{ab} = g_{ba}$).

```
(%i29) decsym(g,0,2,[],[sym(all)]);
```

%i29 Объявляет симметрию для 2х контрвариантных координат тензора g ($g^{ab} = g^{ba}$).

```
(%i30) ishow(1/(16*%pi*G)*((2*L+'R([],[])))*sqrt(-determinant(g)))$
```

%i30 Записывает лагранжиан (показывает результат). *В данном случае под L понимается Λ-член (при выводе потенциала Шварцшильда, впрочем, он далее полагается равным нулю).*

```
(%i31) L0:%,R$
```

%i31 Подставляет в лагранжиан значение компонент скалярной кривизны.

```
(%i32) canform(contract(canform(rename(contract(expand(diff(L0,g([],[m,n]))-
       idiff(diff(L0,g([],[m,n],k)),k)+idiff(rename(idiff(contract(
       diff(L0,g([],[m,n],k,l))),k),1000),l)))))))$
```

%i32 Вычисляет следующее выражение:

$$\frac{\partial L}{\partial g^{\mu\nu}} - \frac{\partial}{\partial x^k}\frac{\partial L}{\partial g^{\mu\nu}_{,k}} + \frac{\partial^2}{\partial x^l \partial x^k}\frac{\partial L}{\partial g^{\mu\nu}_{,k,l}}.$$

что соответствует 10 уравнениям, которые были впервые получены Гильбертом на основе вариационного принципа (3).

```
(\% i33)ishow(e([m,n],[])=canform(%*16*%pi/sqrt(-determinant(g))))$
```

% i33 Присваивает тензору $e_{\mu\nu}$ значение предыдущего выражения, умноженного на $16\pi\sqrt{-\det(g)}$, предварительно его упростив с помощью функции **canform**. Показывает результат (**ishow**).

```
(%i34) EQ:ic_convert(%)$
```

%i34 Присваивает EQ предыдущее выражение, предварительно преобразовав его в выражение, понятное пакету ctensor (т.е. из индексного выражения в компонентное).

```
(%i35) ct_coords:[t,r,u,v];
```

%i35 Переопределяет координаты в виде t,r,u,v (под u и v понимаются координаты θ и φ).

```
(%i36) lg:ident(4);
```

%i36 Определяет метрический тензор в виде единичной матрицы.

```
(%i37) lg[1,1]:B;
       lg[2,2]:-A;
       lg[3,3]:-r^2;
       lg[4,4]:-r^2*sin(u)^2;
```

%i37 Присваивает компонентам метрики следующие значения: $g_{11} = B$, $g_{22} = -A$, $g_{3,3} = -r^2$, $g_{4,4} = -r^2 \sin u^2$.

(Будем искать метрику в виде $ds^2 = Bdt^2 - Adr^2 - r^2 d\theta^2 - r^2 \sin^2\theta d\varphi^2$, где A и B функции только координаты r.

Метрику можно также искать в виде (см., например, [11])

$$ds^2 = e^{2\nu}dt^2 - e^{2\lambda}dr^2 - e^{2\mu}(d\theta^2 + \sin^2\theta d\varphi^2), \qquad (8)$$

где ν, λ и μ функции только одной координаты r. Очевидно, в этом случае $\mu = ln\, r$. Коэфициенты "2"выбраны для удобства дальнейших *вычислений (с тем же успехом можно бы было искать решение в виде $ds^2 = e^{\nu}dt^2 - e^{\lambda}dr^2 - r^2d\Omega^2$, где $d\Omega^2 = d\theta^2 + \sin^2\theta d\varphi^2$). Ковариатные/смешанные компоненты тензора Риччи и тензора Эйнштейна можно получить и вывести на экран в wxMaxima с помощью функций* ***ricci(true)/uricci(true)*** *и* ***leinstein(true)/einstein(true)*** *после того, как задана матрица метрического тензора* ***lg*** *(см, в качестве примера Приложение).*

```
(%i41) kill(dependencies);
```

%i41 Стирает из памяти функциональные зависимости.

```
(%i42) dependencies(A(r),B(r));
```

%i42 Определяет A и B как функции координаты r.

```
(%i43) cmetric();
```

%i43 Рассчитывает обратную матрицу для метрики и подготавливает пакет ctensor для дальнейших вычислений.

```
(%i44) christof(false);
```

%i44 Вычисляет символы Кристоффеля (чтобы не загромождать вычисления, не выводит их на экран: **false**).

```
(%i45) e:zeromatrix(4,4);
```

%i45 Определяет e как нулевую матрицу 4×4.

```
(%i46) ev(EQ);
```

% i46 Вычисляет значения тензора (выражение EQ)с учётом вышеопределенной метрики.

```
(%i47) E:expand(radcan(ug.e));
```

% i47 Присваивает E произведение матриц **ug** (обратная метрическая матрица) и **e** (предыдущее выражение), предварительно его упростив. Функция **radcan** упрощает выражение, содержащее логарифмы, экспоненты и радикалы, приводя его к каноническому виду.

```
(%i48) exp:findde(E,2);
```

%i48 Присваивает **exp** (exp — название выражения) массив дифференциальных уравнений, получаемых из предыдущей матрицы 4×4.

```
(%i49) solve(ode2(exp[1],A,r),A);
```

%i49 Решает обыкновенное дифференциальное уравнение, соответствующее первому элементу массива **exp**, находя A как функцию r: $A(r)$.

```
(%i50) %,%c=-2*M;
```

%i50 Подставляет в предыдущее выражение значение константы, равное $2M$. *Поскольку любая новая теория, претендующая на роль общей, должна включать в себя как предельные случаи уже существующие проверенные теории, из ОТО с необходимостью следуют* *уравнения Ньютона: в пределе $r \to \infty$ должно выполняться:*

$$g_{00} = 1 - \frac{2MG}{c^2 r}, \tag{9}$$

где второе слагаемое соответствует ньютоновскому потенциалу. В естественной системе единиц, гравитационный радиус, $2MG/c^2$, равен $2M$.

```
(%i51) a:%[1],%c=-2*M;
```

%i51 Присваивает функции A решение с учётом константы (в выражение на шаге назад подставляет значение константы).

```
(%i52) ode2(ev(exp[2],a),B,r);
```

%i52 Решает второе обыкновенное дифференциальное уравнение, в которое уже подставлено значение функции A, относительно функции B(r).

```
(%i53) b:ev(%,%c=rhs(solve(rhs(%)*rhs(a)=1,%c)[1]));
```

%i53 Находит вторую константу из условия $A \cdot B = 1$. Подставляет полученное значение в выражение для B, присваивает функции B решение с учётом константы.

На наш взгляд, здесь содержится некоторая непоследовательность. Условие $A \cdot B = 1$ следует непосредственно из уравнений Эйнштейна (это уравнение, точнее даже более общее, $\ln A(t,r) + \ln B(t,r) = f(t)$, получается при сложении первых двух уравнений Эйнштейна). Произвольную функцию времени $f(t)$ выбирают в виде ноля, пользуясь свободой преобразования времени вида $t = f(t')$. Условие $A \cdot B = 1$ в случае выбора метрики в виде (8) запишется, очевидно, в виде $\nu(r) = -\lambda(r)$. Впрочем, вторую константу %i53 мы можем выбрать как раз опираясь на вышеописанную свободу преобразований времени.

```
(%i54) factor(ev(ev(exp[3],a,b),diff));
```

%i54 Упрощает третье дифференциальное уравнение, в которое подставлены A и B как решения предыдущих дифференциальных уравнений. Убеждаемся, что третье уравнение при этом превращается в тождество $0 = 0$.

```
(%i55) lg:ev(lg,a,b),L=0$
```

%i55 Подставляет в метрический тензор полученные значения функций A и B, а $-\Lambda$- член полагает равным нулю.

```
(%i56) ug:invert(lg);
```

%i56 Находит обратную матрицу.

```
(%i57) block([title: "Schwarzschild Potential for Mass M=2",M:2.],
         wxplot3d([r*cos(th),r*sin(th),1-ug[1,1]],[r,5.,50.],[th,-%pi,%pi],
         ['grid,20,30],['z,-2,0],[psfile],['legend,title]));
```

%i57 Заголовок: "Потенциал Шварцшильда для массы M=2. Строит в блоке 3хмерный график $x = r\cos(\theta)$, $y = r\sin(\theta)$, $z = 1 - g_{11}^{-1}$. r

меняется от 5 до 50, θ от $-\pi$ до π, z от -2 до 0, масса $M = 2$. Формируется файл ps, легенда и название показаны на графике.

После того как мы нашли метрику, можно записать выражение, аналогичное уравнению (9), *при этом второе слагаемое будет теперь соответствовать потенциалу Шварцшильда.*

Приложение. Пример использования пакета wxMaxima для иллюстрации вывода метрик Шварцшильда и Рейсснера-Нордстрёма

(%i1) `kill(all);`

(%o0) *done*

(%i1) `load(ctensor);`

(%o1) $C : /PROGRA~1/MAXIMA~1.0/share/maxima/5.30.0/share/tensor/ctensor.mac$

Можно воспользоваться функцией **csetup**:

(%i2) `csetup();`

Enter the dimension of the coordinate system:4;

Do you wish to change the coordinate names?y;

Enter a list containing the names of the coordinates in order[t,r,theta,phi];

Do you want to

1. Enter a new metric?

2. Enter a metric from a file?

3. Approximate a metric with a Taylor series? 1;

Is the matrix 1. Diagonal 2. Symmetric 3. Antisymmetric 4. GeneralAnswer 1, 2, 3 or 4 :

1;

Row 1 Column 1: B;

Row 2 Column 2: -A;

Row 3 Column 3: $-r\hat{}2$;

Row 4 Column 4: $-r\hat{}2 * sin(theta)\hat{}2$;

Matrix entered. Enter functional dependencies with DEPENDS or 'N' if none

depends([A,B],[r]);

Do you wish to see the metric?y; $\begin{pmatrix} B & 0 & 0 & 0 \\ 0 & -A & 0 & 0 \\ 0 & 0 & -r^2 & 0 \\ 0 & 0 & 0 & -r^2\sin(\theta)^2 \end{pmatrix}$

(%o2) *done*

Эти команды можно ввести и вручную:

(%i3) `kill(all);`

(%o0) *done*

(%i1) `load(ctensor);`

(%o1) $C:/PROGRA~1/MAXIMA~1.0/share/maxima/5.30.0/share/tensor/ctensor.mac$

(%i2) `ct_coords:[t,r,theta,phi];`

(%o2) $[t, r, \theta, \phi]$

(%i3) `depends([A,B],[r]);`

(%o3) $[\mathrm{A}(r), \mathrm{B}(r)]$

(%i4) `lg:matrix([B,0,0,0],[0,-A,0,0],[0,0,-r^2,0],[0,0,0,-r^2*sin(theta)^2]);`

(%o4) $$\begin{pmatrix} B & 0 & 0 & 0 \\ 0 & -A & 0 & 0 \\ 0 & 0 & -r^2 & 0 \\ 0 & 0 & 0 & -r^2 \sin(\theta)^2 \end{pmatrix}$$

(%i5) `cmetric(true);`

$Doyouwishtoseethemetricinverse?y;$

(%t5) $$\begin{pmatrix} \frac{1}{B} & 0 & 0 & 0 \\ 0 & -\frac{1}{A} & 0 & 0 \\ 0 & 0 & -\frac{1}{r^2} & 0 \\ 0 & 0 & 0 & -\frac{1}{r^2 \sin(\theta)^2} \end{pmatrix}$$

(%o5) *done*

Символы Кристоффеля:

(%i6) `christof(mcs);`

(%t6) $$mcs_{1,1,2} = \frac{\frac{d}{d\,r}\,B}{2\,A}$$

(%t7) $$mcs_{1,2,1} = \frac{\frac{d}{d\,r}\,B}{2\,B}$$

(%t8) $$mcs_{2,2,2} = \frac{\frac{d}{d\,r}\,A}{2\,A}$$

(%t9) $$mcs_{2,3,3} = \frac{1}{r}$$

(%t10) $$mcs_{2,4,4} = \frac{1}{r}$$

(%t11) $$mcs_{3,3,2} = -\frac{r}{A}$$

(%t12) $$mcs_{3,4,4} = \frac{\cos(\theta)}{\sin(\theta)}$$

(%t13) $$mcs_{4,4,2} = -\frac{r \sin(\theta)^2}{A}$$

(%t14) $$mcs_{4,4,3} = -\cos(\theta)\,\sin(\theta)$$

(%o14) *done*

Ковариантные компоненты тензора Риччи:

(%i15) `ricci(true);`

(%t15) $$ric_{1,1} = \frac{\frac{d^2}{d\,r^2}\,B}{2\,A} - \frac{\left(\frac{d}{d\,r}\,B\right)^2}{4\,A\,B} - \frac{\left(\frac{d}{d\,r}\,A\right)\,\left(\frac{d}{d\,r}\,B\right)}{4\,A^2} + \frac{\frac{d}{d\,r}\,B}{r\,A}$$

(%t16) $$ric_{2,2} = -\frac{\frac{d^2}{d\,r^2}\,B}{2\,B} + \frac{\left(\frac{d}{d\,r}\,B\right)^2}{4\,B^2} + \frac{\left(\frac{d}{d\,r}\,A\right)\,\left(\frac{d}{d\,r}\,B\right)}{4\,A\,B} + \frac{\frac{d}{d\,r}\,A}{r\,A}$$

(%t17) $ric_{3,3}=-\frac{r\,\left(\frac{d}{d\,r}\,B\right)}{2\,A\,B}+\frac{r\,\left(\frac{d}{d\,r}\,A\right)}{2\,A^2}-\frac{1}{A}+1$

(%t18) $ric_{4,4}=-\frac{r\sin\left(\theta\right)^2\,\left(\frac{d}{d\,r}\,B\right)}{2\,A\,B}+\frac{r\sin\left(\theta\right)^2\,\left(\frac{d}{d\,r}\,A\right)}{2\,A^2}-\frac{\sin\left(\theta\right)^2}{A}+$
$\sin\left(\theta\right)^2$

(%o18) *done*

Смешанные компоненты тензора Риччи:

```
(%i19) uricci(true);
```

(%t19) $uric_{1,1}=\frac{2\,r\,A\,B\,\left(\frac{d^2}{d\,r^2}\,B\right)-r\,A\,\left(\frac{d}{d\,r}\,B\right)^2+\left(4\,A-r\,\left(\frac{d}{d\,r}\,A\right)\right)\,B\,\left(\frac{d}{d\,r}\,B\right)}{4\,r\,A^2\,B^2}$

(%t20) $uric_{2,2}=\frac{2\,r\,A\,B\,\left(\frac{d^2}{d\,r^2}\,B\right)-r\,A\,\left(\frac{d}{d\,r}\,B\right)^2-r\,\left(\frac{d}{d\,r}\,A\right)\,B\,\left(\frac{d}{d\,r}\,B\right)-4\,\left(\frac{d}{d\,r}\,A\right)\,B^2}{4\,r\,A^2\,B^2}$

(%t21) $uric_{3,3}=\frac{r\,A\,\left(\frac{d}{d\,r}\,B\right)+\left(-r\,\left(\frac{d}{d\,r}\,A\right)-2\,A^2+2\,A\right)\,B}{2\,r^2\,A^2\,B}$

(%t22) $uric_{4,4}=\frac{r\,A\,\left(\frac{d}{d\,r}\,B\right)+\left(-r\,\left(\frac{d}{d\,r}\,A\right)-2\,A^2+2\,A\right)\,B}{2\,r^2\,A^2\,B}$

(%o22) *done*

Ковариантные компоненты тензора Эйнштейна:

```
(%i23) leinstein(true);
```

(%t23) $lein_{1,1}=\frac{\left(r\,\left(\frac{d}{d\,r}\,A\right)+A^2-A\right)\,B}{r^2\,A^2}$

(%t24) $lein_{2,2}=\frac{r\,\left(\frac{d}{d\,r}\,B\right)+(1-A)\,B}{r^2\,B}$

(%t25) $lein_{3,3}=\frac{r\,\left(2\,r\,A\,B\,\left(\frac{d^2}{d\,r^2}\,B\right)-r\,A\,\left(\frac{d}{d\,r}\,B\right)^2+\left(2\,A-r\,\left(\frac{d}{d\,r}\,A\right)\right)\,B\,\left(\frac{d}{d\,r}\,B\right)-2\,\left(\frac{d}{d\,r}\,A\right)\,B^2\right)}{4\,A^2\,B^2}$

(%t26) $lein_{4,4}=\frac{r\sin\left(\theta\right)^2\,\left(2\,r\,A\,B\,\left(\frac{d^2}{d\,r^2}\,B\right)-r\,A\,\left(\frac{d}{d\,r}\,B\right)^2+\left(2\,A-r\,\left(\frac{d}{d\,r}\,A\right)\right)\,B\,\left(\frac{d}{d\,r}\,B\right)-2\,\left(\frac{d}{d\,r}\,A\right)\,B^2\right)}{4\,A^2\,B^2}$

(%o26) *done*

Смешанные компоненты тензора Эйнштейна:

```
(%i27) einstein(true);
```

(%t27) $ein_{1,1}=\frac{r\,\left(\frac{d}{d\,r}\,A\right)+A^2-A}{r^2\,A^2}$

(%t28) $ein_{2,2}=-\frac{r\,\left(\frac{d}{d\,r}\,B\right)+(1-A)\,B}{r^2\,A\,B}$

(%t29) $ein_{3,3}=-\frac{2\,r\,A\,B\,\left(\frac{d^2}{d\,r^2}\,B\right)-r\,A\,\left(\frac{d}{d\,r}\,B\right)^2+\left(2\,A-r\,\left(\frac{d}{d\,r}\,A\right)\right)\,B\,\left(\frac{d}{d\,r}\,B\right)-2\,\left(\frac{d}{d\,r}\,A\right)\,B^2}{4\,r\,A^2\,B^2}$

(%t30) $ein_{4,4}=-\frac{2\,r\,A\,B\,\left(\frac{d^2}{d\,r^2}\,B\right)-r\,A\,\left(\frac{d}{d\,r}\,B\right)^2+\left(2\,A-r\,\left(\frac{d}{d\,r}\,A\right)\right)\,B\,\left(\frac{d}{d\,r}\,B\right)-2\,\left(\frac{d}{d\,r}\,A\right)\,B^2}{4\,r\,A^2\,B^2}$

(%o30) *done*

Получим статичное сферически-симметричное решение уравнения Эйнштейна без заряда.

```
(%i31) ode2(ev(lein[1,1],theta=%pi/2),A,r);
```

(%o31) $\log\left(A\right)-\log\left(A-1\right)=\log\left(r\right)+\%c$

```
(%i32) logcontract(%);
```

(%o32) $\log\left(\frac{A}{A-1}\right) = \log(r) + \%c$

```
(%i33) solve(%,A);
```

(%o33) $[A = \frac{e^{\%c}\, r}{e^{\%c}\, r - 1}]$

(будем пользоваться граничными условиями для определения констант, при этом для удобства вычислений G=c=1)

```
(%i34) %,%c=-log(2*M);
```

(%o34) $[A = \frac{r}{2\left(\frac{r}{2\,M} - 1\right)\, M}]$

```
(%i35) A:%[1],%c=-log(2*M);
```

(%o35) $A = \frac{r}{2\left(\frac{r}{2\,M} - 1\right)\, M}$

```
(%i36) ode2(ev(lein[2,2],A),B,r);
```

(%o36) $B = \%c\, e^{-2\left(\frac{\log(r)}{2\,M} - \frac{\log(r-2\,M)}{2\,M}\right)\, M}$

```
(%i37) logcontract(%);
```

(%o37) $B = -\frac{\%c\,(2\,M - r)}{r}$

(Используем граничные условия AB=1, откуда найдём константу %c. Затем полученное значение подставим в правую часть уравнения выше, чтобы получить функцию B)

```
(%i38) B:ev(%,%c=rhs(solve(rhs(%)*rhs(A)=1,%c)[1]));
```

(%o38) $B = -\frac{2\,M - r}{r}$

Проверим, удовлетворяют ли полученные функции A и B третьему уравнению:

```
(%i39) factor(ev(ev(lein[3,3],A,B),diff));
```

(%o39) 0

Таким образом, метрический тензор Шварцшильда (1916)):

```
(%i40) lg:ev(lg,A,B);
```

(%o40) $\begin{pmatrix} -\frac{2\,M-r}{r} & 0 & 0 & 0 \\ 0 & -\frac{r}{2\left(\frac{r}{2\,M}-1\right)\, M} & 0 & 0 \\ 0 & 0 & -r^2 & 0 \\ 0 & 0 & 0 & -r^2\,\sin(\theta)^2 \end{pmatrix}$

```
(%i41) ug:invert(lg);
```

(%o41) $\begin{pmatrix} -\frac{r}{2\,M-r} & 0 & 0 & 0 \\ 0 & -\frac{2\left(\frac{r}{2\,M}-1\right)\, M}{r} & 0 & 0 \\ 0 & 0 & -\frac{1}{r^2} & 0 \\ 0 & 0 & 0 & -\frac{1}{r^2\,\sin(\theta)^2} \end{pmatrix}$

Выведем теперь метрику с учётом заряда. Будем считать, что тензор энергии-импульса уже известен.

```
(%i1) kill(all);
```

(%o0) $done$

```
(%i1) load(ctensor);
```

(%o1) $C:/PROGRA\,1/MAXIMA\,1.0/share/maxima/5.30.0/share/tensor/ctensor.mac$

```
(%i2) ct_coords:[t,r,theta,phi];
```

(%o2) $[t, r, \theta, \phi]$

```
(%i3) depends([A,B],[r]);
```

(%o3) $[\mathrm{A}(r), \mathrm{B}(r)]$

```
(%i4) lg:matrix([B,0,0,0],[0,-A,0,0],[0,0,-r^2,0],[0,0,0,-r^2*sin(theta)^2]);
```

(%o4) $$\begin{pmatrix} B & 0 & 0 & 0 \\ 0 & -A & 0 & 0 \\ 0 & 0 & -r^2 & 0 \\ 0 & 0 & 0 & -r^2\sin(\theta)^2 \end{pmatrix}$$

```
(%i5) cmetric();
```

(%o5) $done$

```
(%i6) christof(false);
```

(%o6) $done$

```
(%i7) uricci(false);
```

(%o7) $done$

```
(%i8) einstein(false);
```

(%o8) $done$

```
(%i9) solve(ode2(ein[1,1]=q^2/r^4,A,r),A);
```

(%o9) $$[A = \frac{r^2}{r^2 - \%c\,r + q^2}]$$

```
(%i10) %,%c=2*M;
```

(%o10) $$[A = \frac{r^2}{-2\,r\,M + r^2 + q^2}]$$

```
(%i11) A:%[1],%c=2*M;
```

(%o11) $$A = \frac{r^2}{-2\,r\,M + r^2 + q^2}$$

```
(%i12) solve(ode2(ev(ein[2,2],A)=q^2/r^4,B,r),B);
```

(%o12) $$[B = -\frac{2\,\%c\,r\,M - \%c\,r^2 - \%c\,q^2}{r^2}]$$

(опять же из граничных условий)

```
(%i13) %,%c=1;
```

(%o13) $[B=-\frac{2\,r\,M-r^2-q^2}{r^2}]$

```
(%i14) B:%[1],%c=1;
```

(%o14) $B=-\frac{2\,r\,M-r^2-q^2}{r^2}$

```
(%i15) expand(B);
```

(%o15) $B=-\frac{2\,M}{r}+\frac{q^2}{r^2}+1$

```
(%i16) factor(ev(ev(ein[3,3]+q^2/r^4,A,B),diff));
```

(%o16) 0

Метрика Рейсснера (1916)-Нордстрёма (1918):

```
(%i17) lg:expand(ev(lg,A,B));
```

(%o17) $$\begin{pmatrix} -\frac{2\,M}{r}+\frac{q^2}{r^2}+1 & 0 & 0 & 0 \\ 0 & -\frac{r^2}{-2\,r\,M+r^2+q^2} & 0 & 0 \\ 0 & 0 & -r^2 & 0 \\ 0 & 0 & 0 & -r^2\sin\left(\theta\right)^2 \end{pmatrix}$$

```
(%i18) ug:expand(invert(lg));
```

(%o18) $$\begin{pmatrix} \frac{1}{-\frac{2\,M}{r}+\frac{q^2}{r^2}+1} & 0 & 0 & 0 \\ 0 & \frac{2\,M}{r}-\frac{q^2}{r^2}-1 & 0 & 0 \\ 0 & 0 & -\frac{1}{r^2} & 0 \\ 0 & 0 & 0 & -\frac{1}{r^2\sin(\theta)^2} \end{pmatrix}$$

Список литературы

[1] maxima.sourceforge.net/docs/manual/en/maxima_27.html [дата обращения: 10.07.14]

[2] maxima.sourceforge.net/docs/manual/en/maxima_26.html [дата обращения: 10.07.14]

[3] maxima.sourceforge.net/docs/manual/en/maxima_25.html [дата обращения: 10.07.14]

[4] www.elegio.it/max/demo01/einhil.dem.txt [дата обращения: 10.07.14]

[5] http://www.vttoth.com/CMS/projects/61-the-maxima-computer-algebra-system [дата обращения: 10.07.14]

[6] Winterberg F., Z. Naturforsch, **59 a**, p.715, 2004

[7] Логунов А.А., Мествиришвили М.А., Петров В.А., УФН, 2004, Т. 174, №6

[8] D.Hilbert. die Grundlagen der Physik (Nachr.Ges.Wiss. Gottingen), **3**, 395 (1915); Перевод на русский см. в сборнике «Альберт Эйнштейн и теория гравитации» (М.:Мир, 1979)

[9] Ландау Л. Д., Лифшиц Е. М. Теоретическая физика: Учеб. пособие. в 10 т. Т.II Теория поля. — М.: Наука, 1988, 512 стр.

[10] Хриплович И. Б., Общая теория относительности. — Ижевск: НИЦ «Регулярная и хаотическая динамика», 2001, 120 стр.

[11] Кокарев С. С. Введение в общую теорию относительности. — Ярославль: ЯрГУ, 2010, 368 стр.